
SUPPLEMENTARY MATERIAL 2

Chronometric data and models for archaeobotanical samples
The models in this report were constructed using OxCal 4.3.2 (Bronk Ramsey, 2009) and the IntCal13 calibration curve (Reimer et al., 2013).

Revised phasing scheme
	Phase
	Sub-phase
	Estimated date range (centuries AD)

	1
	1.1
	late ninth – early tenth

	2
	2.1
	mid tenth

	2
	2.2
	mid tenth – late tenth

	2
	2.3
	late tenth – mid eleventh

	3
	3.1
	mid eleventh – mid twelfth

	3
	3.2
	early twelfth – mid twelfth

	3
	3.3
	early thirteenth

	4
	4.1
	thirteenth – fourteenth

	4
	4.2
	thirteenth – sixteenth

	5
	5.1
	sixteenth and later



St Mary’s Grove
Data from Carver 2010
	Feature
	Context
	Material
	Laboratory no.
	Uncalibrated age BP

	F584 (S12)
	2247
	burnt wood and grain
	HAR-7039
	1270 ± 70

	F584 (S12)
	2247
	burnt wood and grain
	HAR-7040
	1120 ± 70

	F584 (S12)
	2247
	burnt wood and grain
	HAR-7041
	1310 ± 90




Data from FeedSax
	Feature
	Context
	Material
	Laboratory no.
	Uncalibrated age BP

	130
	1682
	3 x Avena grain
	OxA-37395
	1111 ± 25

	585
	2223
	3 x Avena grain
	OxA-37533
	930 ± 28

	136
	2136
	3 x Avena grain
	OxA-37534
	1120 ± 27

	136
	2134
	3 x Avena grain
	OxA-37669
	1076 ± 26

	(SFB)
	1988
	3 x Triticum free-threshing grain
	OxA-37535
OxA-37536
	910 ± 27
888 ± 27

	449
	2178
	3 x Secale cereale grain
	OxA-37537
	973 ± 28

	471
	1929
	3 x Triticum free-threshing grain
	OxA-37538
	896 ± 28

	435
	2102 066/077
	3 x Secale cereale grain
	OxA-37539
	962 ± 27

	435
	2102 066/079
	3 x Secale cereale grain
	OxA-37540
	919 ± 26

	435
	2102 069/078
	3 x Secale cereale grain
	OxA-37541
	981 ± 27



OxCal model in CQL2
Plot()
 {
  Outlier_Model("Charcoal",Exp(1,-10,0),U(0,3),"t");
  Sequence("St Marys Grove") {
  Boundary("Start whole");
  Phase("whole")
  {
   Phase("North-East area")
   {
    Sequence("ovens to pits sequence")
    {
     Boundary ("Start Period 3");
     Phase ("Period 3")
     {
      Phase("Oven F130")
      {
       R_Date("OxA-37395", 1111, 25);
      };
      Phase("Oven F584")
      {
       R_Date("HAR-7039", 1270, 70)
       {
        Outlier("Charcoal",1);
       };
       R_Date("HAR-7040", 1120, 70)
       {
        Outlier("Charcoal",1);
       };
       R_Date("HAR-7041", 1310, 90)
       {
        Outlier("Charcoal",1);
       };
      };
      Phase("Oven F585")
      {
       R_Date("OxA-37533", 930, 28);
      };
      Phase("Pit F136")
      {
       Sequence("F136 pit fills")
       {
        Boundary("Start 2136");
        Phase("context 2136")
        {
         R_Date("OxA-37534", 1120, 27);
        };
        Boundary("end 2136 start 2134");
        Phase("context 2134")
        {
         R_Date("OxA-37669", 1076, 26);
        };
        Boundary("end 2134");
       };
      };
     };
     Boundary("end Period 3 start Period 4");
     Phase ("Period 4")
     {
      R_Date("OxA-37537", 973, 28);
     };
     Boundary("End Period 4");
    };
   };
   Phase("Central area")
   {
    Phase("F517 SFB")
    {
     R_Date("OxA-37535", 910, 27);
     R_Date("OxA-37536", 888, 27);
    };
    Phase("F435 quarry pit")
    {
     R_Date("OxA-37539", 962, 27);
     R_Date("OxA-37540", 919, 26);
     R_Date("OxA-37541", 981, 27);
    };
   };
   Phase("South-West area")
   { 
    Phase("F471")
    {
     R_Date("OxA-37538", 896, 28);
    }; 
   };
  };
  Boundary("End whole");
  };
 };

Bath Street
Data from Carver 2010
	Feature
	Context
	Material
	Laboratory no.
	Uncalibrated age BP

	Posthole F58
	1082
	Charcoal (oak heartwood and hazel)
	HAR-5291
	1320 ± 70



Data from FeedSax
	Feature
	Context
	Sample
	Material
	Laboratory no.
	Uncalibrated age BP

	Pit F227
	1069B
	IIB
	3 x Secale cereale grain
	OxA-37630
	938 ± 26

	Pit F227
	1071A
	IA
	3 x Secale cereale grain
	OxA-37790
	949 ± 25



Modelling was not deemed necessary for the interpretation of these data.

Tipping Street
Data from Dodd et al. 2014
	Feature
	Context
	Sample
	Material
	Laboratory no.
	Uncalibrated age BP

	Kiln 3408
	3203
	205
	Triticum sp. grain
	SUERC-38774
	1160 ± 30

	
	
	
	Alnus glutinosa charcoal fragment
	SUERC-38773
	1230 ± 30

	Kiln 4301
	3201
	203
	Triticum sp. grain
	SUERC-38778
	1105 ± 30

	
	
	
	Corylus avellana charcoal fragment
	SUERC-38779
	1130 ± 30

	Kiln 4287
	4338
	407
	Triticum sp. grain
	SUERC-38997
	1160 ± 30

	
	4304
	403
	Triticum sp. grain
	SUERC-38780
	1100 ± 30



Data from FeedSax
	Feature
	Context
	Sample
	Material
	Laboratory no.
	Uncalibrated age BP

	Kiln 3408
	3203
	205
	3 x Triticum free-threshing grains
	OxA-37625
	1082 ± 25

	
	3205
	204
	3 x Triticum free-threshing grains
	OxA-37626
	1159 ± 26

	Kiln 3401
	3201
	203
	3 x Triticum free-threshing grains
	OxA-37627
	1121 ± 26

	Kiln 4287
	4319
	406
	3 x Triticum free-threshing grains
	OxA-37628
	1109 ± 26

	Pit 366
	381
	11
	3 x Secale cereale grains
	OxA-37629
	1110 ± 24



OxCal model in CQL2
Plot()
 {
  Outlier_Model("Charcoal",Exp(1,-10,0),U(0,3),"t");
  Sequence("whole")
  {
   Boundary("Start Phase 2");
   Phase("Phase 2")
   {
    Phase ("Context 381")
    {
     R_Date("OxA-37629", 1110, 24);
    };
    Sequence ("Area 3 kilns")
    {
     Boundary("Start kiln 3408 use");
     Phase ("Kiln 3408 use")
     {
      R_Date("SUERC-38774", 1160, 30);
      R_Date("SUERC-38773", 1230, 30)
      {
       Outlier("Charcoal",1);
      };
      R_Date("OxA-37625", 1082, 25);
     };
     Boundary("End kiln 3408 use start kiln 3408 demolition");
     Phase("Kiln 3408 demolition")
     {
      R_Date("OxA-37626", 1159, 26);
     };
     Boundary("End kiln 3408 demolition start kiln 3401 use");
     Phase("Kiln 3401 use")
     {
      R_Date("SUERC-38778", 1105, 30);
      R_Date("SUERC-38779", 1130, 30)
      {
       Outlier("Charcoal",1);
      };
      R_Date("OxA-37627", 1121, 26);
     };
     Boundary("End kiln 3401 use");
    };
    Sequence("Kiln 4287")
    {
     Boundary("Start kiln 4287 use");
     Phase("Kiln 4287")
     {
      R_Date("SUERC-38997", 1160, 30);
      R_Date("SUERC-38780", 1100, 30);
      R_Date("OxA-37628", 1109, 26);
     };
     Boundary("End kiln 4287");
    };
   };
   Boundary("End Phase 2");
  };
 };

